Pokémon and Graph Theory
Introduction
Pokémon has hugely influenced adults and children worldwide. The video game is a cornerstone of
popular culture and brings together hundreds of thousands of players, each of which are looking to
outwit the other and gain an edge.
But not many of those players know that Maple can help you become a better pokémon player.
This application will use concepts from graph theory to determine which the best routes a player
can take in the game to accomplish its two main objectives: to Catch Em'All® and become a
Pokémon Master.
We will focus on the role-playing game Pokémon Blue released in 1998 by Nintendo. The
application only considers pokémons from Generation I, a reduced group of the hundreds of
pokémon existing in modern Pokémon games.
In the game, the player is a young boy/girl that travels around the region of Kanto, catching
different animal-like creatures called pokémon and engaging in pokémon battles. The player's
objective is to form a team of six pokémon to challenge into battle other players (or npcs) and win
against them.
To catch pokémon, the player needs to search for them in Routes. Routes are numbered places in
the game where the user can encounter wild pokémon to catch. Each pokémon has a certain
probability of being encountered by the player while walking in tall grass, fishing or surfing.
We will explore the answers of the following questions:
1. What is the most common/rare wild pokémon to catch?
2. Given a certain desired team, what is the shortest path among routes/cities/towns I should
take to catch them?
3. If I'm looking for a great team, in which Route should I be spending the most time on?
The data used to model the graphs has been obtained from the well known fan-driven pokemon
database and wiki Bulbapedia.

How we will organize the data:
A graph is a visual representation of a list of connections between vertices through the use of edges,
with the optional addition of a weight.
Vertices: We will have two types of vertices: Routes and pokémon.
Edges: A pokémon will be connected to a Route if there exists the possibility of it appearing on the
wild in that route.
Weights: Each edge will be given a weight that represents the probability of that pokémon
appearing on the Route to which it is connected by that edge.

Generating the Graphs
We will write a matrix containing the folowing:
Pidgey

Route 1

0.55

where 0.55 represents the probability of Pidgey appearing in Route 1. If the probability is 0 then the
pokémon does not appear in the wild on this Route and so there will be no edge connecting the
pokémon and the Route.
Notice that there are three ways of encountering pokémon in the wild: in tall grass (or similar),
surfing or fishing. Furthermore, for some pokémon in some Routes, the probability for finding them
varies depending on which method is used to encounter it. For example, in Route 12, Magikarp has
a probability of 0.25 or 1 of appearing when fishing, depending on which fishing rod the player uses.
To simplify the graph, we will calculate the average for these different probabilities and work with
the result as one general probability for the pokémon in that area.
We will generate two graphs: the first one to visualize the connections between pokémons and
Routes, the second one to visualize the adjacency of Routes and how to get from one to the other.
To start, we call the necessary packages for graph modelling.
> with(GraphTheory):
with(Statistics):
with(ColorTools):
with(ArrayTools):
with(LinearAlgebra):
We import the edges for the first and second graphs.
> graph_edges_pokemon := ImportMatrix("this://graph_edges_pokemon1.csv");

>
graph_edges_routes := ImportMatrix("this://graph_edges_routes1.csv");

Setting styles for the graphs
When using Maple's DrawGraph command, we may want to choose the colors for the vertices,
the font or the amount of space between them. We can also do the same for edges. In this
section we will create four different styles of graph layouts to use in the sections below.
> pokemon_graph_style := [
vertexcolor = "DarkRed"
,vertexfontcolor = white
,vertexborder = false

>

,edgethickness = 0
,edgecolor
= "Black"
,vertexshape = "stadium"
,vertexfont = ["Arial", 20]]:
routes_highlight_style := [
color = "DarkCyan"
,fontcolor = white
,border = false
,shape = "stadium"
,font = ["Arial", 20]
]:
water_routes := ["Route 12", "Route 14", "Route 19", "Route 20", "Route 21", "Route 24"]:
water_routes_style := [
color = "navy"
,fontcolor = white
,border = false
,shape = "stadium"
,font = [Arial, 12]]:
caves := ["Cerulean Cave", "Diglett's Cave", "Mt. Moon", "Rock Tunnel", "Seafoam Islands",
"Victory Road's Cave", "Power Plant"]:
caves_style := [
color = brown
,fontcolor = white
,border = false
,shape = "stadium"
,font = [Arial, 12]]:
routes_graph_style := [
vertexcolor = "OliveDrab"
,vertexfontcolor = white
,vertexborder = false
,edgethickness = 0
,edgecolor
= black
,vertexshape = "stadium"
,vertexfont = [Arial, 12]]:
neighbors_graph_style := [
vertexcolor = Color("Purple")
,vertexfontcolor = white
,vertexborder = false

,edgethickness = 0
,edgecolor
= Color("RGB",[191/255, 191/255, 191/255])
,vertexshape = "stadium"
,vertexfont = [Arial, 16]
,weightfont = [Arial, 12]]:

Graph for paths between routes
We will use a similar approach as before to graph the connections between routes, which in
simple words, is the map of the game.
We first write the edges in set form, to use them later in the Graph command.
> routes_edges := {}:
for i from 1 to 45 do
ith_row := Row(graph_edges_routes, i);
routes_edges:= {op(routes_edges), {ith_row(1), ith_row(2)}};
routes_edges := remove(x -> evalb(x = {}), routes_edges);
end do:
We use the Graph function from the GraphTheory package.
> routes_graph := Graph(routes_edges):
routes_list := Vertices(routes_graph):
To draw the graph, we will use the DrawGraph function from the GraphTheory package.

> HighlightVertex(routes_graph, water_routes, stylesheet = water_routes_style);
HighlightVertex(routes_graph, caves, stylesheet = caves_style);
DrawGraph(routes_graph, size = [1000, 1000], style = spring, stylesheet =
routes_graph_style);

Where:
The green vertices are Routes that are mostly made up of tall grass or where the player will
need a pokémon that knows the move cut to travel through.
The blue vertices are the Routes that are mostly made up of water or where the player will
need a pokémon that knows the move surf to travel through.
The brown vertices are caves or dungeons.

Graph for pokémon and routes
We first write the edges in set form, to use them later in the Graph command from the
GraphTheory toolbox.
> pokemon_edges := {}:
for i from 1 to 241 do
ith_row := Row(graph_edges_pokemon, i);
if(ith_row(2) <> "Diglett's Cave") then
pokemon_edges:= {op(pokemon_edges), [{ith_row(1), ith_row(2)}, ith_row(3)]};
end if;
end do:

We use the Graph function from the GraphTheory toolbox.
> pokemon_graph := Graph(pokemon_edges, weighted):
To draw the graph, we will use the DrawGraph function from the GraphTheory toolbox.
> HighlightVertex(pokemon_graph, remove(x -> x = "Diglett's Cave", routes_list), stylesheet =
routes_highlight_style);
DrawGraph(pokemon_graph, size = [2000, 2000], showweights = false, style = spring,
stylesheet = pokemon_graph_style);

Since the pokémons Diglett and Dugtrio can only be found in Diglett's Cave, they are
disconnnected from the rest of the graph. We therefore add them now:

> pokemon_graph := AddVertex(pokemon_graph, ["Diglett", "Dugtrio", "Diglett's Cave"]):
pokemon_graph := AddEdge(pokemon_graph, {[{"Diglett", "Diglett's Cave"}, 0.95], [
{"Dugtrio", "Diglett's Cave"}, 0.05]}):
Where the red vertices are pokémon and the cyan vertices are places.

Answering some basic questions
By only looking at the graph we can immediately answer the following questions:

What are the routes a pokémon appears in? / Which pokémon can I
find on a given route?
To answer this question, we have to focus on the pokémon's/route's neighborhood.
To get the routes in which we can find Pidgey, we will use Maple's neighborhood function. The
weights shown are the probabilities for finding the pokémon in the given routes.
> my_pokemon := "Pidgey":
pokemon_neighbors := Neighborhood(pokemon_graph, my_pokemon):
pokemon_neighbors_graph := InducedSubgraph(pokemon_graph, [op
(pokemon_neighbors), my_pokemon]):
HighlightVertex(pokemon_neighbors_graph, my_pokemon, inplace = true, stylesheet =
[color = "Navy", shape = "ellipse", fontcolor = "White", font = [Arial, 16]]):
DrawGraph(pokemon_neighbors_graph, size = [500, 500], style = spring, stylesheet =
neighbors_graph_style, showweights = true);

To get the pokémon that we can find in Route 20, we will use a similar approach:
> my_route := "Route 20":
route_neighbors := Neighborhood(pokemon_graph, my_route):
route_neighbors_graph := InducedSubgraph(pokemon_graph, [op(route_neighbors),
my_route]):
HighlightVertex(route_neighbors_graph, my_route, inplace = true, stylesheet = [color =
"Navy", shape = "stadium", fontcolor = "White", font = [Arial, 20]]):
DrawGraph(route_neighbors_graph, size = [1000, 700], style = bipartite, thickness = 0,
showweights = true, stylesheet = neighbors_graph_style);

What is the most common/rare pokémon?
To answer this question we will consider both the probabilities of the pokémon appearing and
the number of places it appears in. Since each pokémon is connected to the routes it appears in,
we can count them by counting the pokémon's adjacency matrix and weight matrix, which will
give us the number of routes and probabilities for each pokémon in the graph in matrix form.
> adjacency := AdjacencyMatrix(pokemon_graph):
weights := WeightMatrix(pokemon_graph):
vertex_ordered_list := convert(pokemon_graph, list)[3]:
To calculate the number of neighbors each pokémon has, and the sum of the probabilities we
add the columns for the pokémon of each matrix.

> pokemon_inMatrix := remove(x -> member(x, routes_list), Vertices(pokemon_graph)):
total_pokemon := nops(pokemon_inMatrix):
num_of_neighbors := [seq(nops(Neighborhood(pokemon_graph, pokemon_inMatrix[i])), i=
1..total_pokemon)]:
> weights_sum := [seq(add(map(x -> GetEdgeWeight(pokemon_graph, {x, pokemon_inMatrix
[i]}), Neighborhood(pokemon_graph, pokemon_inMatrix[i]))), i=1..total_pokemon)]:
> value_to_consider := [seq([pokemon_inMatrix[i], num_of_neighbors[i]*weights_sum[i]], i=
1..total_pokemon)]:
value_to_consider := sort(value_to_consider,(x, y) -> x[2] < y[2]):
BarChart([seq((value_to_consider[i])[1] = (value_to_consider[i])[2], i = 1..75)], size = [1000,
1000], style = surface, title = "How common is each pokémon?", titlefont = [Arial, 20], font
= [Arial, 10], axesfont = [Arial, 10], labels = ["Multiplied value of total probability and
number of places of appeareance",""], labelfont = [Arial, 16], axes = frame, size = [800,
1000], color = "Navy");

What is the route with most different pokémon?
We are now concerned to the number of pokémon that are connected to a route. This is called
the degree of the route. We therefore need to find the route with the highest degree.
> num_of_routes := nops(routes_list):
most_pokemon_routes := [seq([Degree(pokemon_graph, routes_list[i]), routes_list[i]], i =
1..num_of_routes)]:
most_pokemon_routes := sort(most_pokemon_routes,(x, y) -> x[1] < y[1]):
BarChart([seq((most_pokemon_routes[i])[2] = (most_pokemon_routes[i])[1], i = 1..
num_of_routes)], size = [800,800], style = surface, title = "Routes by number of different
pokémon species", titlefont = [Arial, 20], font = [Arial, 10], axesfont = [Arial, 10], labels =
["Number of different pokémon species",""], labelfont = [Arial, 16], axes = frame, size =
[800,600], color = "Gold");

Routes to visit to catch a desired team
Let's say the player wants to renew the team he/she has. We will make a list of 6 pokémon the
player wants to catch. In this case, it will be:

> to_catch := ["Snorlax", "Pidgey", "Caterpie", "Abra", "Tentacool", "Magikarp"]:
Feel free to change the pokémons' names if you want a different team. We will find the shortest
path between routes to visit and also the path with the best probabilities for each pokémon. We
assume that this is post-game when the player has all routes unlocked and a pokémon that knows
fly. The move fly lets the player travel from one distance to another without having to walk through
connecting routes, it acts as a "teleport" option in the game.

Shortest Path
We must try every possible permutation of the names in order to choose the shortest possible
path.
> all_possible := combinat:-permute(to_catch):
total_possible := nops(all_possible):
shortest_path := routes_list:
for i from 1 to total_possible do
to_catch := all_possible[i];
possible_shortest := []:
team_num := nops(to_catch);
for k from 1 to team_num - 1 do
possible_shortest := [op(possible_shortest), op(ShortestPath(pokemon_graph, to_catch
[k], to_catch[k+1]))];
end do;
possible_shortest := {op(select(x -> member(x, routes_list), possible_shortest))};
if nops(possible_shortest) <= nops(shortest_path) then
shortest_path := possible_shortest;
end if;
end do:
So the minimal sequence of routes the player needs to visit to catch the desired team is:
> shortest_path;

Best Probabilities Path
If we want to consider the walk with best probablities for each pokémon, we may have to visit
some more Routes. We will first check to which Routes the pokémon are connected and select
the one with best probablity. The routes that are connected to a certain pokémon are called the
pokémon's neighbors.
> best_probabilities_path := []:
for j from 1 to 6 do
pokemon := to_catch[j];

>

best_edge := [];
#Any weight is less than 2 since we are working with probabilities.
best_weight := 2;
neighbors_for_pokemon := Neighbors(pokemon_graph, pokemon);
total_neighbors := nops(neighbors_for_pokemon);
for r from 1 to total_neighbors do
neighbor := neighbors_for_pokemon[r];
weight := GetEdgeWeight(pokemon_graph, {neighbor, pokemon});
if weight <= best_weight then
best_edge := [neighbor, pokemon];
best_weight := weight;
end if;
end do:
best_probabilities_path := [op(best_probabilities_path), best_edge];
end do:
So the routes to visit with best probabilities for each pokemons is:
> best_probabilities_path;

If we only wish to have the routes we get:
> routes_only := {op(map(x -> (best_probabilities_path[x])[1], [seq(1..6)]))};

